
1 Identifiers

〈word〉 ::= 〈Letter〉 { [’ | - ] 〈Letter〉 }

〈identifier〉 ::= 〈word〉 { 〈word〉 }

〈java-identifier〉 ::= 〈word〉 { 〈word〉 }

〈list-of-identifiers〉 ::= 〈identifier〉
| 〈identifier〉 [,] and 〈identifier〉
| 〈identifier〉 , 〈list-of-identifiers〉

2 Literals

〈number〉 ::= zero | one | ... | twelve | 〈Digit〉 { 〈Digit〉 }

〈ordinal-literal〉 ::= [the] (first | ... | twelfth | 〈number〉 (st|nd|rd|th))

〈string-literal〉 ::= 〈Double-Quote〉 { 〈Non-Quote-Character〉 { 〈Non-Quote-Character〉
} | 〈Single-Quote〉 〈expression〉 〈Single-Quote〉 } 〈Double-Quote〉

〈literal〉 ::= 〈number〉 | 〈string-literal〉

3 Expressions

〈simple-expression〉 ::= 〈literal〉 | 〈identifier〉

〈list-expression〉 ::= 〈simple-expression〉 [,] and 〈simple-expression〉
| 〈simple-expression〉 , 〈list-expression〉

〈expression〉 ::= 〈list-expression〉
| only 〈simple-expression〉
| 〈simple-expression〉

〈argument-list〉 ::= 〈simple-expression〉
| 〈simple-expression〉 [,] and 〈simple-expression〉
| 〈simple-expression〉 , 〈argument-list〉

〈ordinal-expression〉 ::= 〈ordinal-literal〉 page
| page numbered by 〈identifier〉

4 Auxiliary words

〈does〉 ::= does | do | did

〈is〉 ::= is | are | was | were

〈has〉 ::= has | have | had

〈is-or-likes〉 ::= is | are | likes | like

〈pronoun〉 ::= he | she | it | they

〈pronoun-o〉 ::= him | her | it | them

5 Conditions

〈simple-condition〉 ::= 〈expression〉 〈conditional-operator〉 〈expression〉
| (everypony | everything) in 〈expression〉 〈conditional-operator〉 〈expression〉
| (anypony | anything) in 〈expression〉 〈conditional-operator〉 〈expression〉
| 〈expression〉 〈is〉 (nothing | nopony)
| 〈expression〉 〈is〉 (something | somepony)

〈conditional-operator〉 ::= 〈has〉 less than

| 〈has〉 more than

| 〈has〉 not less than

| 〈has〉 not more than

| 〈is〉 equal to

| 〈is〉 not equal to

| 〈is〉 ([an] element | elements) of
| 〈is〉 not ([an] element | elements) of

6 Punctuation

〈inline-comment〉 ::= , because { 〈Almost-Anything〉 }

〈sentence-end〉 ::= [〈inline-comment〉] (.|!)

〈header-end〉 ::= [〈inline-comment〉] (.|:)

〈question-end〉 ::= [〈inline-comment〉] (?)

7 Statements

〈increment-s〉 ::= 〈identifier〉 got 〈number〉 (more |less|fewer) 〈sentence-end〉

〈assignment-s〉 ::= Did you know [that] 〈identifier〉 〈is-or-likes〉 〈expression〉
〈question-end〉

〈if-s〉 ::= When 〈condition〉 〈header-end〉
{ 〈statement〉 } [In the end, I did this instead 〈header-end〉
{ 〈statement〉 } ] That’s what I did 〈sentence-end〉

〈while-s〉 ::= I did this while 〈condition〉 〈header-end〉
{ 〈statement〉 } [ It didn’t work, but I knew why 〈header-end〉
{ 〈statement〉 } ] [ In the end, I did this instead 〈header-end〉
{ 〈statement〉 } ]
That’s what I did 〈sentence-end〉

〈repeat-s〉 ::= I did this [〈number〉 times|once] 〈header-end〉
{ 〈statement〉 } [ It didn’t work, but I knew why 〈header-end〉
{ 〈statement〉 } ] [ In the end, I did this instead 〈header-end〉
{ 〈statement〉 } ]
That’s what I did 〈sentence-end〉

〈global-declaration-s〉 ::= Yes, [I mean] that 〈identifier〉 〈sentence-end〉

〈call-function-s〉 ::= I [also] (did|made|caused) 〈identifier〉 of 〈argument-list〉
〈sentence-end〉

〈call-function-each-s〉 ::= I [also] (did|made|caused) 〈identifier〉 of each[of]
〈argument-list〉 〈sentence-end〉

〈call-function-and-assign-s〉 ::= 〈identifier〉 (did|made) 〈identifier〉
of 〈argument-list〉 〈sentence-end〉

〈call-function-each-and-assign-s〉 ::= 〈identifier〉 (did|made) 〈identifier〉
of each[of] 〈argument-list〉 〈sentence-end〉

〈java-class-import-s〉 ::= I enchanted 〈identifier〉 with 〈java-identifier〉 〈sentence-end〉

〈java-constructor-call-s〉 ::= I woke up 〈identifier〉
with 〈identifier〉 and 〈argument-list〉 〈sentence-end〉

〈java-method-call-s〉 ::= [I told 〈identifier〉] I asked [〈identifier〉]
〈java-method-invocation〉 〈sentence-end〉

〈java-method-invocation〉 ::= about 〈argument-list〉 (so|if|what|how) [〈pronoun〉]
〈java-identifier〉

| about 〈identifier〉 made 〈java-identifier〉
| about 〈identifier〉 with〈argument-list〉 of 〈java-identifier〉
| (what|if) 〈identifier〉 [with 〈argument-list〉] (〈is〉|〈has〉) 〈java-identifier〉

〈java-field-set-s〉 ::= I (gave|sold) 〈expression〉 to 〈simple-expression〉 of

〈identifier〉 〈sentence-end〉

〈java-field-get-s〉 ::= I (took|got|stole) 〈simple-expression〉 of 〈identifier〉
and [I] gave 〈pronoun-o〉 to 〈identifier〉 〈sentence-end〉

〈print-s〉 ::= I [quickly] (said|sang|wrote) [that|,|:]
〈expression〉 〈sentence-end〉

〈comment-s〉 ::= By the way, 〈Almost-Anything〉 〈sentence-end〉

〈array-assignment-s〉 ::= On 〈ordinal-expression〉 of [the] [book] 〈identifier〉
I (wrote|scribbled|noted) [about|what I knew about] 〈expression〉 〈sentence-end〉

〈array-retrieval-s〉 ::= On 〈ordinal-expression〉 of [the] [book] 〈identifier〉
I read about 〈identifier〉 〈sentence-end〉

〈array-init-s〉 ::= [Today] I found a book (named|titled) 〈identifier〉 [today]
〈sentence-end〉

〈statement〉 ::= 〈array-init-s〉|〈array-retrieval-s〉|〈array-assignment-s〉
| 〈comment-s〉|〈print-s〉
| 〈java-field-get-s〉|〈java-field-set-s〉
| 〈java-method-call-s〉|〈java-constructor-call-s〉|〈java-class-import-s〉
| 〈call-function-each-and-assign-s〉|〈call-function-and-assign-s〉
| 〈call-function-each-s〉|〈call-function-s〉
| 〈global-declaration-s〉|〈repeat-s〉|〈while-s〉|〈if-s〉
| 〈assignment-s〉|〈increment-s〉

8 Functions

〈function〉 ::= I learned [about] 〈identifier〉 [with 〈list-of-identifiers〉] 〈header-end〉
{〈statement〉}
That’s about 〈identifier〉 [with 〈identifier〉] 〈sentence-end〉

〈main-function〉 ::= Today I learned [[about] 〈identifier〉] 〈header-end〉
{〈statement〉}
Your faithful student, 〈identifier〉 [〈sentence-end〉]

9 Modules

〈c-style-module〉 ::= Dear Princess Celestia: 〈identifier〉 〈header-end〉
{ 〈function〉 }
〈main-function〉

〈wrapped-module〉 ::= Dear Princess Celestia: 〈identifier〉 〈header-end〉
Today I learned [about] 〈identifier〉 〈header-end〉
{ 〈function〉 }
Your faithful student, 〈identifier〉 [〈sentence-end〉]

〈program〉 ::= 〈c-style-module〉 | 〈wrapped-module〉

